package kgame5.kgui.gameui.digi {
import kgame5.kutil.Util;

import flash.display.MovieClip;
import flash.display.Sprite;
import flash.geom.Rectangle;

/**
 * @author kk
 * 一个数字
 * 这样使用动态连接是为了避免自己手动制作mc_d
 */
public class ADigi {
	private var 
		mc:Sprite,
			mc_strip:MovieClip;
	private var cellWidth:int;
	
	private var rect:Rectangle;
	
	//sLinkID是条带的名字
	public function ADigi(sLinkID:String){
		this.mc=new Sprite();
		
		mc_strip=Util.AttachMC(sLinkID);
		cellWidth=mc_strip.width/10;
		mc.addChild(mc_strip);
		
		rect=new Rectangle(0,0,cellWidth,mc_strip.height);
		mc.scrollRect=rect;
		
	}
	public function getMC():Sprite{
		return mc;
	}
	public function show(value:int){
		mc.visible=true;
		
		rect.x=cellWidth*value;
		mc.scrollRect=rect;
	}
	public function hide(){
		mc.visible=false;
	}
	public function move(x:int){
		mc.x=x;
		//mc.y=y;
	}
	
	//进行一下便宜，这个是为了对其用的
	public function ofsX(p_ofsX:int){
		mc.x-=p_ofsX;
	}
}
}
